

```
; ctcss_modificeret_OZ1LN_03b.asm
; Original SW: CTCSS Encoder v1.4 for 16c84 Tony Hunt VK5AH 14-4-99
; Changed to PIC 16F628 and adopted for use in Storno CQM5000 (OZ1LN)
; Enabled internal pull-up 31-08-2020
```

```
; Use 4MHz Crystal
```

```
*****
```

```
;
;          -----U-----
; TONE OUTPUT----- RA2 |1      18| RA1-----
; ----- RA3 |2      17| RA0-----
; ----- RA4 |3      16| OSC1--XTAL
; +5V 22k/ICSP Enable ---- MCLR |4      15| OSC2--XTAL
; Ground ----- Vss |5      14| VDD-- +5 V
; TONE SEL----- RB0 |6      13| RB7-- ICSP
; TONE SEL----- RB1 |7      12| RB6-- ICSP
; TONE SEL----- RB2 |8      11| RB5-----PTT LOW
; TONE SEL----- RB3 |9      10| RB4-----TONE SEL
```

```
;
;          -----
;          PIC16F628
;*****
```

```
list p=16f628 ;list directive to define processor
#include <p16f628.inc> ;processor specific variable definitions
__CONFIG _CP_OFF & _LVP_OFF & _BODEN_OFF & _MCLRE_OFF & _WDT_OFF &
_PWRTE_ON & _XT_OSC
```

```
CNT1 EQU 20H ;PRESCALLER COUNTER REG 1
CNT2 EQU 21H ;PRESCALLER COUNTER REG 2
CNT3 EQU 22H ;PRESCALLER COUNTER REG 3
TONREG EQU 23H ;REGISTER FOR TONE NUMBER
```

```
PRES1 EQU 24H ;PRESCALLER STORE 1
PRES2 EQU 25H ;PRESCALLER STORE 2
PRES3 EQU 26H ;PRESCALLER STORE 3
```

```
PC EQU 2
```

```
ORG 0x000 ;processor reset vector
goto MAIN
```

```
MAIN
```

```
MOVLW 0x07 ;Turn comparators off and
MOVWF CMCON ;enable pins for I/O
```

```
BSF STATUS,RP0 ;SELECT REG BANK 1
BCF TRISA,2 ;SETUP RA2 AS OUTPUT FOR TONE
BCF OPTION_REG,NOT_RBPU ;ENABLE PULLUP RESISTORS ON PORTB
MOVLW 0xFF ;SET PORTB AS I/Ps
MOVWF TRISB ;RB0-7 all inputs
```

```
BCF STATUS,RP0 ;SELECT REG BANK 0
BCF INTCON,GIE ;DISABLE GIE INTERRUPTS
BSF INTCON,RBIE ;ENABLE PORT B CHANGE INTERRUPT
```

```
START
```

```
MOVF PORTB,W ;READ DIP SWITCHES INTO W
MOVWF TONREG ;STORE IN TONREG
COMF TONREG,f ;Invert
BCF TONREG,5 ;CLEAR BITS 5,6&7 AS
BCF TONREG,6 ;THEY ARE IRRELEVANT
BCF TONREG,7
```

```
MOVF TONREG,W ;RE-STORE INTO W
ADDWF TONREG,F ;MULTIPLY BY 3
ADDWF TONREG,F ;FOR TONE TABLE OFFSET
```

```

MOVF    TONREG,W           ;RELOAD TO W FOR OFFSET
CALL    TONTBL             ;GOTO TONE LOOKUP TABLE
MOVWF   PRES1              ;STORE FIRST VALUE FROM TABLE

INCF    TONREG,F           ;INCREMENT FOR NEXT TABLE VALUE
MOVF    TONREG,W           ;RELOAD W FOR OFFSET
CALL    TONTBL             ;GOTO TONE LOOKUP TABLE
MOVWF   PRES2              ;STORE SECOND VALUE FROM TABLE

INCF    TONREG,F           ;INCREMENT FOR NEXT TABLE VALUE
MOVF    TONREG,W           ;RELOAD W FOR OFFSET
CALL    TONTBL             ;GOTO TONE LOOKUP TABLE
MOVWF   PRES3              ;STORE THIRD VALUE FROM TABLE

RELOAD
MOVF    PRES1,W            ;LOAD W WITH PRESET FOR 1
MOVWF   CNT1               ;PRELOAD PRESCALLER 1
MOVF    PRES2,W            ;LOAD W WITH PRESET FOR 3
MOVWF   CNT3               ;PRELOAD PRESCALLER 3
MOVF    PRES3,W            ;LOAD W WITH PRESET FOR 2 READY

PTTCHK                                ;Check PTT is low
BTFSC   PORTB,5              ;TEST PTT LINE
GOTO    WAIT                 ;SLEEP TILL LOW

OK      CALL    CHORUS

OUTPUT
BTFSS   PORTA,2              ;TEST O/P HI OR LO AT PRESENT
GOTO    HI                   ;IF LO GO SET HI
GOTO    LO                   ;IF HI GO SET LO

HI
NOP                                           ;BALANCE OUT CYCLE TIME MARK&SPACE
BSF     PORTA,2              ;MAKE O/P HI
GOTO    RELOAD               ;RETURN TO START OF CYCLE

LO
BCF     PORTA,2              ;MAKE O/P LO
GOTO    RELOAD               ;RETURN TO START OF CYCLE

CHORUS
PRE1
DECFSZ  CNT1,F               ;DECREMENT PRESCALLER 1
GOTO    PRE2
GOTO    FINE                 ;CALL FINE ADJUST SUBROUTINE

PRE2
MOVWF   CNT2                 ;RELOAD PRESCALLER 2

PREE
DECFSZ  CNT2,F               ;DECREMENT PRESCALLER 2
GOTO    PREE                 ;GO BACK AND DECREMENT PRE2 TILL 00
GOTO    PRE1                 ;GO BACK AND DECREMENT PRE1 TILL 00

FINE
DECFSZ  CNT3,F               ;DECREMENT PRESCALLER 3
GOTO    FINE                 ;GO BACK AND DECREMENT PRE3 TILL 00
RETURN  ;RETURN TO CHANGE O/P AND CHECK PTT ETC

WAIT
BCF     INTCON,RBIF          ;CLEAR PORT B CHANGE INT FLAG
SLEEP   ;SLEEP AND WAIT FOR PTT
NOP
BCF     INTCON,GIE          ;DISABLE GLOBAL INTERRUPTS AGAIN
GOTO    START                ;Reload switch settings from inputs and
start again

```

TONTBL

ADDWF PC,F ; W+PC ->PC, JUMP DOWN TABLE

```
; "1" i tabel betyder stel (LO) på RBx 16f628 PIN
;
; TONE FREQ Posn RB4 RB3 RB2 RB1 RB0
;
DT 86,82,10 ; 000 67.0 0 0 0 0 0
DT 84,80,0x0F ; 001 71.9 0 0 0 0 1
DT 7F,85,0x0F ; 002 74.4 0 0 0 1 0
DT 7B,7C,0x0F ; 003 77.0 0 0 0 1 1
DT 76,86,0x0F ; 004 79.7 0 0 1 0 0
DT 72,82,0x0F ; 005 82.5 0 0 1 0 1
DT 74,8F,0x0E ; 006 85.4 0 0 1 1 0
DT 63,91,10 ; 007 88.5 0 0 1 1 1
DT 5F,9A,10 ; 008 91.5 0 1 0 0 0
DT 5B,47,11 ; 009 94.8 0 1 0 0 1
DT 55,88,11 ; 010 97.4 0 1 0 1 0
DT 56,49,11 ; 011 100.0 0 1 0 1 1
DT 09,6E,0xB9 ; 012 103.5 0 1 1 0 0
DT 08,81,0xC9 ; 013 107.2 0 1 1 0 1
DT 0A,71,98 ; 014 110.9 0 1 1 1 0
DT 0B,76,83 ; 015 114.8 0 1 1 1 1
DT 0A,0xCA,83 ; 016 118.8 1 0 0 0 0
DT 0A,9A,83 ; 017 123.0 1 0 0 0 1
DT 09,0xF1,83 ; 018 127.3 1 0 0 1 0
DT 08,93,9D ; 019 131.8 1 0 0 1 1
DT 07,94,0xB0 ; 020 136.5 1 0 1 0 0
DT 07,7C,0xAD ; 021 141.3 1 0 1 0 1
DT 07,85,0xA5 ; 022 146.2 1 0 1 1 0
DT 08,7B,89 ; 023 151.4 1 0 1 1 1
DT 08,80,83 ; 024 156.7 1 1 0 0 0
DT 08,47,86 ; 025 162.2 1 1 0 0 1
DT 08,86,78 ; 026 167.9 1 1 0 1 0
DT 07,90,85 ; 027 173.8 1 1 0 1 1
DT 09,5A,66 ; 028 179.9 1 1 1 0 0
DT 07,9E,78 ; 029 186.2 1 1 1 0 1
DT 02,01,0x9D ; 030 1Khz TEST 1 1 1 1 0
DT 03,15,20 ; 031 1750 Eu Tone 1 1 1 1 1
```

ORG 2100H

DE "TonyHunt VK5AH CTCSSENC v 1.4 modific.by OZ1LN"

END

All CTCSS tones are within EIA standard of .08% .

Use a 4Mhz crystal with 2x18pF capacitors.

PTT line is active LOW. PTT is RB7 Pin 13.

OUTPUT tone is on RA2 Pin1.

All RB port pins with internal pullup resistors and will float HI if left open. The binary inputs are read each time the PTT is keyed.